Various Mathieu functions

B. Schnizer

schnizer@itp.tu-graz.ac.at

```
Needs["Graphics`Graphics`"]

diff[aa_, qq_] = D[w[z], {z, 2}] + (aa - 2 qq Cos[2 z]) w[z] == 0

(aa - 2 qq Cos[2 z]) w[z] + w''[z] == 0
```

The values for q and a must be written with a decimal point in function calls for Mathieu functions !

```
xo = 2 π + 2; vq = 2.;

prl = {Hue[0], Line[{{π, -1.1}, {π, 1.1}}], Line[{{2 π, -1.1}, {2 π, 1.1}}]};
```

Basically periodic solutions and their non-periodic partner

Even periodic solutions

```
ta = Table[MathieuCharacteristicA[r, vq], {r, 0, 3}]

{-1.51396, 2.3792, 5.17267, 9.37032}
```

$\{a_0, \quad a_1, \quad a_2, \quad a_3\}$

From Matematica program, $ce_r(a_r, q, z)$ is basically periodic

```
tc = Table[Plot[MathieuC[ta[[k]], vq, z] , {z, 0, xo},
   Ticks → {PiScale, Automatic}, Epilog -> prl, DisplayFunction → Identity,
   PlotLabel → SequenceForm["q = ", vq, ", r = ", k - 1, ", ",
     Subscript["a", k - 1], " = ", ta[[k]]]], {k, Length[ta]}];
```

```
Show[GraphicsArray[{{tc[[1]], tc[[2]]}, {tc[[3]], tc[[4]]}}], ImageSize → 500];
```



■ **From Matematica program. According to theory, second solution is n o t basically periodic**

```
tcs = Table[Plot[MathieuS[ta[[k]], vq, z] / MathieuSPrime[ta[[k]], vq, 0] , {z, 0, xo},
   Ticks → {PiScale, Automatic}, Epilog -> prl, DisplayFunction → Identity,
   PlotLabel → SequenceForm["q = ", vq, ", r = ", k - 1, ", ",
     Subscript["a", k - 1], " = ", ta[[k]]]], {k, Length[ta]}];

Show[GraphicsArray[{{tcs[[1]], tcs[[2]]}, {tcs[[3]], tcs[[4]]}}], ImageSize → 500];
```




```
Table[
 MathieuS[ta[[k]], vq, π / 4] / MathieuSPrime[ta[[k]], vq, 0] , {k, 4}]

{-0.104266, 0.373546, 0.251556, 0.268811}
```

```
Table[
 MathieuS[ta[[k]], vq, 5 π / 4] / MathieuSPrime[ta[[k]], vq, 0] (-1) ^ (k - 1) , {k, 4}]
```

{-0.104266, 0.373546, 0.251556, 0.268811}

```
% - %%
```

$\{1.66533\times 10^{-16}, 1.66533\times 10^{-16}, 5.55112\times 10^{-17}, 2.77556\times 10^{-16}\}$

■ Corresponding odd solution from numeric integration

```
tfc = Table[NDSolve[{diff[ta[[k]], vq], w[0] == 0, w'[0] == 1},
     w, {z, 0, xo }], {k, Length[ta]}] // Flatten;

uc = Table[Plot[w[z] /. tfc[[k]], {z, 0, xo}, DisplayFunction → Identity,
    PlotLabel → SequenceForm["q = ", vq, ", a = ", ta[[k]]],
    Ticks → {PiScale, Automatic}], {k, Length[ta]}];

Show[GraphicsArray[{{uc[[1]], uc[[2]]}, {uc[[3]], uc[[4]]}}], ImageSize → 500];
```




```
w[0] /. tfc[[4]]
```

1.27751×10^{-21}

```
w[π] /. tfc[[4]]
```

-0.0542987

```
w[π / 4] /. tfc[[4]]
```

0.345556

```
w[5 π / 4] /. tfc[[4]]
```

-0.327803

So even the solution corresponding to a = 9.37032 is n o t periodic though one might guess this from the curve.

■ Odd periodic solutions

■ From Matematica program, $se_r(a_r, q, z)$ is basically periodic

```
tb = Table[MathieuCharacteristicB[r, vq], {r, 4}]
```

```
{-1.39068, 3.67223, 9.14063, 16.1277}
```

```
ts = Table[Plot[MathieuS[tb[[k]], vq, z] , {z, 0, xo}, Ticks → {PiScale, Automatic},
   Epilog -> prl, DisplayFunction  → Identity, PlotPoints → 100,
   PlotLabel → SequenceForm["q = ", 1, ", r = ", k, ", ", Subscript["b", k], " = ", tb[[k]]]],
  {k, Length[tb]}];
```

```
Show[GraphicsArray[{{ts[[1]], ts[[2]]}, {ts[[3]], ts[[4]]}}], ImageSize → 500];
```



■ From Matematica program. According to theory, second solution is n o t basically periodic

```
tsc = Table[Plot[MathieuC[tb[[k]], vq, z] / MathieuC[tb[[k]], vq, 0] , {z, 0, xo}, Ticks →
    {PiScale, Automatic}, Epilog -> prl, DisplayFunction  → Identity, PlotPoints → 100,
   PlotLabel → SequenceForm["q = ", 1, ", r = ", k, ", ", Subscript["b", k], " = ", tb[[k]]]],
  {k, Length[tb]}];
```

```
Show[GraphicsArray[{{tsc[[1]], tsc[[2]]}, {tsc[[3]], tsc[[4]]}}], ImageSize → 500];
```




```
Table[Plot[MathieuC[tb[[k]], vq, z] / MathieuC[tb[[k]], vq, 0] , {z, 0, xo},

tfsc = Table[NDSolve[{diff[tb[[k]], vq], w[0] == 1, w'[0] == 0},
     w, {z, 0, xo } ], {k, Length[tb]}] // Flatten;

psc = Table[Plot[w[z] /. tfsc[[k]], {z, 0, xo }, DisplayFunction → Identity,
    PlotLabel → SequenceForm["q = ", vq, ", a = ", tb[[k]]],
    Ticks → {PiScale, Automatic}], {k, Length[tfsc]}];

Show[GraphicsArray[{{psc[[1]], psc[[2]]}, {psc[[3]], psc[[4]]}}], ImageSize → 500];
```



Periodic, but not basically periodic (stable) solutions

```
vq
```

```
2.
```

```
ap = 7.5;

ppc = Plot[MathieuC[7.5, 2., z] , {z, 0, 2 xo}, Ticks → {PiScale, Automatic},
  DisplayFunction → Identity];
pps = Plot[MathieuS[7.5, 2., z] , {z, 0, 2 xo}, Ticks → {PiScale, Automatic},
  DisplayFunction → Identity];

Show[GraphicsArray[{ppc, pps}], ImageSize → 500];
```






```
tpc = NDSolve[{diff[7.5, 2], w[0] == 1, w'[0] == 0},
    w, {z, 0, 2 xo }] // Flatten;
tps = NDSolve[{diff[7.5, 2], w[0] == 0, w'[0] == 1},
    w, {z, 0, 2 xo }] // Flatten;

plc = Plot[w[z] /. tpc, {z, 0, 2 xo}, DisplayFunction → Identity,
   PlotLabel → SequenceForm["q = ", vq, ", a = ", ap],
   Ticks → {PiScale, Automatic}];
pls = Plot[w[z] /. tps, {z, 0, 2 xo}, DisplayFunction → Identity,
   PlotLabel → SequenceForm["q = ", vq, ", a = ", ap],
   Ticks → {PiScale, Automatic}];

Show[GraphicsArray[{plc, pls}], ImageSize → 500];
```





Unperiodic (unstable) solutions

```
Table[{z, MathieuC[5., 4., z]}, {z, 0, xo}]
```

```
{{0, 0.615037 - 0.751016 ⅈ}, {1, 0.740595 - 0.904333 ⅈ}, {2, -0.544352 + 0.664703 ⅈ},
 {3, 1.59833 - 1.9517 ⅈ}, {4, 5.59224 - 6.82863 ⅈ}, {5, -4.87019 + 5.94694 ⅈ},
 {6, 7.63238 - 9.31982 ⅈ}, {7, 34.4467 - 42.0625 ⅈ}, {8, -29.1392 + 35.5815 ⅈ}}
```

```
Table[{z, MathieuS[5., 4., z]}, {z, 0, xo}]
```

```
{{0, 0. + 0. ⅈ}, {1, 1.0501 + 0.859966 ⅈ}, {2, -0.891741 - 0.730283 ⅈ},
 {3, 1.80145 + 1.47528 ⅈ}, {4, 6.83218 + 5.59515 ⅈ}, {5, -5.96301 - 4.88335 ⅈ},
 {6, 9.28993 + 7.6079 ⅈ}, {7, 42.06 + 34.4446 ⅈ}, {8, -35.5804 - 29.1382 ⅈ}}
```

```
xo = 15.; iq = 4.; ia = 5.; soic = NDSolve[{diff[5., iq], w[0] == 1, w'[0] == 0},
   w, {z, 0, xo }] // Flatten;
sois = NDSolve[{diff[5., iq], w[0] == 0, w'[0] == 1},
    w, {z, 0, xo }] // Flatten;

pic = Plot[w[z] /. soic, {z, 0, xo}, DisplayFunction → Identity,
   PlotLabel → SequenceForm["q = ", iq, ", a = ", ia],
   Ticks → {PiScale, Automatic}];
pis = Plot[w[z] /. sois, {z, 0, xo}, DisplayFunction → Identity,
   PlotLabel → SequenceForm["q = ", iq, ", a = ", ia],
   Ticks → {PiScale, Automatic}];

picd = Show[pic, PlotRange → {{0, π}, 2 {-1, 1}}];
pisd = Show[pis, PlotRange → {{0, π}, 2 {-1, 1}}];

Show[GraphicsArray[{{pic, pis}, {picd, pisd}}], ImageSize → 500];
```